Communications Between the Server, Language and Other Applications

The sound synthesis server has a predefined specification for the messages it can respond to. This is detailed in the file: 

Server-Command-Reference

also linked from the Main Help. You might want to read the following help files in combination with the material below:

ClientVsServer	
NodeMessaging	
ServerTiming	

There is a direct consequence of this interface: the Server can be controlled from Java or C executables, or live from Perl or Python, or indeed, any interpreted programming language or compiled executable that can send the right instructions via Open Sound Control (OSC). Alternative language extensions built to support common SuperCollider constructs like creating SynthDefs exist, for instance Haskell and Scheme libraries written by Rohan Drape, or the Java-based JCollider (http://www.sciss.de/jcollider/) or ScalaCollider (http://www.sciss.de/scalaCollider/index.html) by Hanns Holger Rutz. 

It's very important to remember that objects of type Bus, Buffer, Synth and Server itself are just abstractions on the language side meant to represent what's going on on the Server. They make things simpler for you to code, avoiding low level messaging syntax. But when you manipulate them, under the surface you are causing OSC messages to be sent over to the Server. 
 
The interesting part is getting messages back from the Server to the client (the Language). This can be acheived in two ways; via SendTrig/SendReply and an OSCresponder, or via a Bus or Buffer. In the former, the rate of messages is set on the server; in the latter, you poll the server from the language. Both are asynchronous; you'll have an older value by the time the message reaches the language.  

For a reminder of Open Sound Control in SuperCollider, and the OSCresponder class, see the file 

10.1 Open Sound Control.rtf  

in the Computer Music course (http://www.cogs.susx.ac.uk/users/nc81/courses/cm1/workshop.html) 

Remember that there are 128 audio buses and 4096 control buses by default; these were detailed in the

6.3 Buses.rtf

and

8.2 Control Busses.rtf 

lectures. There are also 1024 Buffers; these are only used rarely to pass data back to the language, either by writing a sound file which is then read by the language side SoundFile class, or by the get methods of Buffer. This is usually because a specialist UGen has been set up that way, and you may not have any cause to use this yourselves. 




To give an example, let's show two ways of getting data from a MouseX UGen back to the language. 

//method 1: using SendTrig and OSCresponder

{SendTrig.kr(Impulse.kr(1), 78, MouseX.kr)}.play //impulse forces message sending once per second

(
OSCresponder(s.addr,'/tr',{ arg time,responder,msg;

if(msg[2] ==78,{  }); 

"received mouse position!".postln;
Post << msg <<nl;  //msg[3] is just the value

}).add;	
)



By the way, to send multiple pieces of information at once, with your own chosen message tags, SendReply is very useful. The following example demonstrates this with random number generation from LFNoise0 UGens, but the same principle would work for MouseX and MouseY values at the same time, or some collection of machine listening UGens extracting multiple features of a signal (we'll need this later in the course).  

//using SendReply and OSCresponder
(
{
	SendReply.kr(Impulse.kr(3), 'sendreplydemo', LFNoise0.kr([3,3]), 1905);
}.play(s);
)

o.remove; o = OSCresponder(nil, 'sendreplydemo', { |t, r, msg| msg.postln }).add;




//method 2: using a Bus

b= Bus.control(s,1);

{Out.kr(b.index, MouseX.kr)}.play

b.get({|val| val.postln;}); //poll once; pass the action function for what you'll do when the message arrives (it takes a short time, this is asynchronous)


There is a special synchronous (immediate) way to communicate quickly between the Internal Server (NOT the separate localhost) and the Language, using shared control buses (there are 1024 of these by default).

[SharedIn]
[SharedOut]

a= {SharedOut.kr(1,MouseX.kr); SinOsc.ar(SharedIn.kr(0,1)) }.play

(
t= Task({

10.do{
s.setSharedControl(0, rrand(100,400));
s.getSharedControl(1).postln;

1.0.wait;
}

}).play
);


b= s.getSharedControl(0); //get the value into the language

b //prove it



Inter application communication is usually effected using OSC, though you could also try sending MIDI messages or flat TCP. See:

The OSC lecture from the original Computer Music course as above
[NetAddr]
[MIDIOut]

Examples will be provided of inter application communication in this course when we discuss audiovisual mappings, since a common need is to sync up a visual application and an audio one.  
 
